

4.6 NONPULSATILE CARDIOVASCULAR MODELS

The frequencies in human cardiovascular pulsations tend to be harmonics of a fundamental heart rate of approximately 1.2 beats per second; healthy adults at rest may have rates ranging from about 0.8 to 1.5 beats per second, and harmonics up to the tenth are important if waveforms near the ventricular outputs are of interest. Exercise or illness may increase the heart rate significantly, and sometimes modelers may have to consider bandwidths of 40 or 50 Hz. These frequencies are much higher than those of concern in the kinetics of most pharmaceutical substances. Thus, for example, most anesthetic agents and muscle relaxants function in the body with their fastest time constants in minutes. Since the pulsatility seems to be added to the slower changes in blood flow, schemes for nonpulsatile cardiovascular modeling have been developed (Rideout-83b, Möller-83, Peskin-79). These methods appear to be particularly useful in multiple models for pharmacokinetic studies, especially in anesthesiology. The dynamic model described here is based on a fourth-order nonpulsatile model (Rideout-83b, pp. 156–157).

Figure 4.6.1a shows the ventricular compliance curves of maximum (systolic) slope and minimum (diastolic) slope; a typical pressure-flow

locus is also shown. This diagram is of key importance in setting up a nonpulsatile cardiovascular model, because it shows that the stroke volume for a ventricle is

$$Q_{SV} = P_{ED} * C_D - P_{ES} * C_S \quad (4.6.1)$$

where C_D and C_S are the diastolic and maximum systolic compliance of the myocardium and P_{ED} is the end-diastolic and P_{ES} the end-systolic pressure, ignoring some nonlinearities which will be given some consideration later (Sunagawa-81, Tham-88). This equation, if multiplied by heart rate H , gives the average outflow of either ventricle:

$$F = (C_D * H) * P_{ED} - (C_S * H) * P_{ES} \quad (4.6.2)$$

Factors K_d and K_s may be introduced, which are such that average atrial pressure is related to end-diastolic pressure by

$$P_{at} = P_{ED} / K_d \quad (4.6.3)$$

and average arterial pressure is similarly related to end-systolic pressure by

$$P_{art} = P_{ES} / K_s \quad (4.6.4)$$

The ventricular outflow, from (4.6.2), (4.6.3), and (4.6.4), is

$$F = G_{pre} * P_{at} - G_{after} * P_{art} \quad (4.6.5)$$

where the preload and afterload conductances are given by

$$\begin{aligned} G_{pre} &= C_D * H * K_d \\ G_{after} &= C_S * H * K_s \end{aligned} \quad (4.6.6)$$




Figure 4.6.1. (a) Ventricular pressure-volume locus, related to minimum and maximum ventricular stiffness (or reciprocal compliance). (b) Venous and arterial pressure-volume relationships.

Figure 4.6.1b shows that the arterial or venous segmental compliance C , together with unstressed volume Q_U , gives total average volume in a segment as

$$q = q_U + p * C \quad (4.6.7)$$

where P is the average pressure. Here q may vary as a result of changes in any of the quantities q_U , p , and/or C . Note that a nonlinear relationship may be used in place of (4.6.7).

If the preload and afterload conductances for the left ventricle are called G_1 and G_2 , and those for the right ventricle G_3 and G_4 , then the ventricular nonpulsatile flow equations may be written as

$$\begin{aligned} G_1 * p_L - G_2 * p_S &= f_L \\ G_3 * p_R - G_4 * p_P &= f_R \end{aligned} \quad (4.6.8)$$

where lowercase letters (p and f) indicate the variables, and uppercase letters the quantities that are ordinarily constant. Also, subscripts L and R have been chosen to indicate the left and right atria, and S and P the systemic and pulmonary arteries.

The systemic and pulmonary equations expressing the pressure drops (principally in capillary beds) are

$$\begin{aligned} p_S - p_R &= R_S * f_S \\ p_P - p_L &= R_P * f_P \end{aligned} \quad (4.6.9)$$

where R_P is the pulmonary peripheral resistance and f_P the total pulmonary flow; R_S and f_S are the corresponding systemic quantities, but may need to be modified if a number of parallel paths are to be considered, as will be shown below.

The volumes q_S , q_R , q_P , and q_L associated with each compliance are given by the integrals of inflow and outflow as shown in (4.6.10). Also, the compliances C_S (total arterial), C_R (total systemic venous), and corresponding pulmonary compliances C_P and C_L may be used with the four volumes (see (4.6.7)) to obtain pressures, as shown below:

$$\begin{aligned} q_S &= \int_0^t (f_L - f_S) dt & p_S &= (q_S - Q_{SU}) / C_S \\ q_R &= \int_0^t (f_S - f_R + f_I) dt & p_R &= (q_R - Q_{RU}) / C_R \\ q_P &= \int_0^t (f_R - f_P) dt & p_P &= (q_P - Q_{PU}) / C_P \\ q_L &= \int_0^t (f_P - f_L) dt & p_L &= (q_L - Q_{LU}) / C_P \end{aligned} \quad (4.6.10)$$

where an infusion flow of blood or plasma, f_i , into the systemic veins is assumed. The 12 equations—(4.6.8), (4.6.9), and (4.6.10)—serve to define a nonpulsatile model, shown in Fig. 4.6.2.

Several parallel systemic paths may need to be included in Fig. 4.6.2, corresponding to capillary beds that run through different kinds of tissue (Tham-90). If there are three such paths, for example, with flow resistances R_{S1} , R_{S2} , and R_{S3} , then the total peripheral systemic resistance used in the first equation of (4.6.9) will be

$$R_S = 1/(1/R_{S1} + 1/R_{S2} + 1/R_{S3}) \quad (4.6.11)$$

The individual flows are given by equations of the form

$$f_{S1} = f_S * R_S/R_{S1} \quad (4.6.12)$$

It should also be noted that if the total volume q_T of the system in Fig. 4.6.2,

$$q_T = q_S + q_R + q_P + q_L \quad (4.6.13)$$

is fixed, with no flow paths to “ground” or to some other system, then there is an “excess integration” in the equations of (4.6.10). This extra integration can be shown to be present because only three integrator commands are needed (to find, for example, q_R , q_P , and q_L), with the remaining blood volume determined from (4.6.13), using

$$q_S = q_T - q_R - q_P - q_L \quad (4.6.14)$$




Figure 4.6.2. Basic non-pulsatile model, with two parallel systemic paths shown.

If four integrations are used, a drift problem may result, with Q_T gradually increasing or decreasing, because of slight inaccuracies in the integrations. However, note that in the following program an excess integration was used without errors appearing; this will usually be possible when digital computation is used. Also, in real physiological systems, blood volume or fluid balance is maintained by the thirst reflex, which governs water intake, and the kidneys, which tend to remove more water when increased volume causes increased system pressure. If such fluid balance equations are included in this model, then they will control Q_T , and four integrations must then be used or the four volumes associated with the four compliances.

The system described by (4.6.8), (4.6.9), and (4.6.10) may be set up in the ACSL program PF-NP. This program includes a means for changing the left ventricular stiffness; here an increase of a factor of two (obtained by decreasing G_2) is shown, and may be moved into the time range of the program by changing TCH to a value such as TCH = 1.

An infusion flow f_i is provided here, in the form of a pulse of amplitude A = 25.0 ml, starting at TSTT = 1.0 sec, and lasting for WID = 2.0 sec; it is not set to repeat within the time range of the problem, but could be by reducing PER.

```

PROGRAM PF-NP
DYNAMIC
  Constant CINT= 0.1
  Constant TF=10.
DERIVATIVE
  Algorithm IALG = 4 $ 'Runge Kutta 2'
  Maxterval MAXT =.05
  Nsteps NSTP = 1

  'Generate LV afterload stiffness G2'
  Constant G2IC=.821, G2NEW=.4105,TCH=1.E6
  Z= T - TCH
  G2= FCNSW(Z,G2IC,G2IC,G2NEW)

  'Flow equations'
  Constant G1=24.,G3=40.,G4=3.889
  FL= G1*PL-G2*PS
  FR= G3*PR-G4*PP
  Constant RS=1.0111,RP=.12222
  FS= (PS-PR)/RS
  FR= (PP-PL)/RP

  'Generate infusion flow FI at STT with period . . .
  PER, width WID, and amplitude A'

```

```

Constant A=25.,TSTT=1.,PER=1.E6,WID=2.
FI=A*PULSE(TSTT,PER,WID)

'Differential Equations'
Constant QSIC=1000.,QRIC=5400.,...
QPIC=500.,QLIC=1800.
QS=INTEG (FL-FS, QSIC)
QR=INTEG (FS-FR+FI, QRIC)
QP=INTEG (FR-FP, QPIC)
QL=INTEG (FP-FL, QLIC)

'Check total volume'
QT=QS+QR+QP+QL

'Find Pressures'
Constant CS=2.6316,CR=225.0,CP=6.9444,CL=42.857
Constant QSU=750.,QRU=4500.,QPU=375.,QLU=1500.
PS=(QS-QSU)/CS
PR=(QR-QRU)/CR
PP=(QP-QPU)/CP
PL=(QL-QLU)/CL
END $ 'of Deriv.'
TERMT (T,GE,TF)
END $ 'Of Dynamic'
END $ 'Of Program'

```

The units used in this example are the so-called "medical" units, with Q in ml, F in ml/s, and P in mmHg. In these units resistances are in mmHg*sec/ml, and compliances in ml/mmHg.

This program was first run as shown (with infusion FI , but with no change in $G2$) for $TF = 10$ sec (see Fig. 4.6.3). The total infusion (25 ml/s for 2 s) was 60 ml, and this amount of change in QT would be expected during the period from $T = 1$ to $T = 3$ sec.

The program was also run with FI zero during the operating period, but with $G2$ decreased (corresponding to an increase in left ventricular stiffness) at $T = 1$ sec (see Fig. 4.6.4). This results in increased PS and QS ; PR and QR also increase, but somewhat more slowly. As a result of these increases, the variables PP , QP , PL , and QL decrease. The flow, after transients settle out, will again be equal, but will be larger because of the stronger left heart. Total volume QT will be constant.

Many changes may be made in the model in addition to the introduction of a number of parallel paths in the systemic circulation (See Eqns. 4.6.11 and 4.6.12) and similar introduction of parallel paths in the pulmonary circulation. Some of these are nonlinear changes, such as the introduction of a square-law term in the diastolic compliance, and nonlinearities in the peripheral resistances.




Figure 4.6.3. Response of the non-pulsatile model of Fig. 4.6.2 to an infusion pulse, FI .

(a) Input FI and total volume QT . Note that the change in QT is linear and occurs only during the infusion.

(b) Volumes associated with the four compliances. All volumes tend to increase with a total which equals 50 ml. Volume QR shows the fastest response, and has the most overshoot.

(c) The varying parts of pressures PR and PS have the same forms as QR and QS , as would be expected.

(d) The flows all tend to be equal, initially, and after transient excursions which are slower as we go from FR through FP , FL and FS , they are again equal.




Figure 4.6.4. Effects of a sudden decrease in G_2 (doubling of left ventricular stiffness) in the nonpulsatile model.
 (a) Blood flows; note that the left heart has a sudden increase in inflow (FS) at $t = 1$ second.
 (b) Pulmonary pressures PP and PL.
 (c) Systemic pressures PS and PR. Volumes QS and QR will have similar waveforms.

The transients shown in the nonpulsatile model solutions indicate that the human CV system has time constants that are not much greater than a typical pulse period (of about 0.8 sec). It was deemed important, therefore, that comparisons be made to check the nonpulsatile model against the corresponding responses for a pulsatile model. This was done, and the nonpulsatile model outputs were found to follow closely the pulsatile outputs, when the latter were averaged over each pulse (Rideout-88).

It is possible to find some closed-form expressions that give the changes in the steady-state conditions in the linear nonpulsatile model, by solving the set of algebraic equations for this system that results when derivative terms are set to zero (Rideout-83b). This may be done using a nonnumeric language such as REDUCE (Hearn-73) or MAXSYMA. Some interesting results are as follows:

1. For a change Δq in total blood volume, the change Δf in cardiac output will be

$$\Delta f = (G_1 G_3 - G_2 G_4) \Delta q / (-\det) \quad (4.6.15)$$

where

$$-\det = (C_L + C_P)[G_3(1 + G_2 R_S) + G_2] + (C_R + C_S)[G_1(1 + G_4 R_P) + G_4] + G_1 G_3 (R_P + R_S C_S) + G_2 G_4 (R_P C_L + R_S C_R) \quad (4.6.16)$$

2. For a change ΔR_S in total systemic peripheral resistance, the change in cardiac output will be

$$\Delta f = -F[G_1 G_3 C_S + G_2 G_4 C_R + G_2 G_3 (C_L + C_P)] \Delta R_S / (-\det) \quad (4.6.17)$$